Symbiosis mediated by sanctions

Supplement to Yoder and Tiffin, “Sanctions, recognition, and variation in mutualistic symbiosis”

Terms

This creates a stand-alone model of symbiosis mediated only by host sanctions against non-cooperating symbionts.

Assume infinite populations of interacting, haploid, hosts and symbionts. A host’s ability to sanction a non-cooperative symbiont is determined by its genotype at a single locus; hosts carrying the allele H at this locus are able to sanction, while those carrying the h allele are not. A haploid symbiont is cooperative if it carries the M allele, but not if it carries the m allele. Hosts and symbionts encounter each other at random and initiate symbiosis without respect to their genotypes.

In successful interaction, the host pays a cost C_H of symbiosis and receives a benefit B_H from a cooperative symbiont; the symbionts pay a cost C_S and receive benefit C_S. Hosts interacting with non-mutualist symbionts (m genotype) pay the cost without gaining the benefit. unless they are able to sanction (H genotype). The effectiveness of sanctions, ω, determines how much of the cost of symbiosis a host can avoid when interacting with a non-cooperative symbiont; greater values of ω mean that sanctions do more to reduce the cost of hosting non-cooperators.

Symbiosis outcomes and fitness

First, a host payoff matrix describes the outcomes of encounters between host genotypes H or h (rows) and symbiont genotypes M and m (columns).

```
In[5]:= pmH = {{BH - CH, - (1 - ω) CH},
           { BH - CH, -CH}};
        MatrixForm[pmH]
```

Out[6]//MatrixForm=

$$\begin{pmatrix} B_H - C_H & (-1+\omega)\, C_H \\ B_H - C_H & -C_H \end{pmatrix}$$

Then, a symbiont payoff matrix (same orientation, for convenience).

```
In[7]:= pmS = {{BS - CS, (1 - ω) BS},
           { BS - CS, BS}};
        MatrixForm[pmS]
```

Out[8]//MatrixForm=

$$\begin{pmatrix} B_S - C_S & (1-\omega)\, B_S \\ B_S - C_S & B_S \end{pmatrix}$$

Fitness expressions

From the payout matrices, we can derive fitness statements for each host and symbiont genotype. For each species, fitness is determined as 1+P, where P is the payout from the symbiosis, determined by the frequencies of the other species’ genotypes.

Host fitness

In[9]:=
```
wH = 1 + pM pmH[[1, 1]] + (1 - pM) pmH[[1, 2]];
wh = 1 + pM pmH[[2, 1]] + (1 - pM) pmH[[2, 2]];
wbarH = pH wH + (1 - pH) wh;
```

Sanctioning hosts have greater fitness than non-sanctioners whenever $\omega C_H(1-p_M)>0$. This requires that $C_H>0$ (there is a cost to hosting), that $\omega>0$ (sanctions are able to reduce the cost of hosting), and $1-p_M>0$ (there are non-cooperative symbionts present in the population).

In[13]:=
```
Simplify[wH - wh]
```

Out[13]= $-\omega\, C_H\, (-1+p_M)$

Symbiont fitnesses

In[14]:=
```
wM  =  1 + pH pmS[[1, 1]] + (1 - pH) pmS[[2, 1]];
wm  =  1 + pH pmS[[1, 2]] + (1 - pH) pmS[[2, 2]];
wbarM  = pM wM + (1 - pM) wm;
```

Non-cooperative symbionts have greater fitness than cooperators whenever $C_S/(\omega B_S)>\omega p_H B_S$. This requires that the cost of symbiosis C_S be greater than the benefits of symbiosis B_S weighted by the effectiveness of sanctions ω and the frequency of sanctioning hosts p_H — that is, the probability that a non-cooperator will encounter a sanctioning host, and the degree to which sanctions can prevent non-cooperators from enjoying the benefit of symbiosis.

In[18]:=
```
FullSimplify[wm - wM]
```

Out[18]= $C_S - \omega\, B_S\, p_H$

Allele frequency dynamics

From the above fitness expressions, we can derive per-generation change in the frequency of host and symbiont alleles, following common expressions.

Host

In[23]:=
```
pH ' = pH wH / wbarH;
ΔpH  = FullSimplify[ pH ' - pH /. ϵ → 1]
```

Out[24]= $$-\frac{\omega\, C_H\, (-1+p_H)\, p_H\, (-1+p_M)}{-1+C_H\, (1+\omega\, p_H\, (-1+p_M))-B_H\, p_M}$$

Symbionts

In[25]:=
```
pM ' = pM wM / wbarM;
ΔpM = FullSimplify[pM ' - pM /. ϵ → 1]
```

Out[26]= $$\frac{(C_S-\omega\, B_S\, p_H)\, (-1+p_M)\, p_M}{1+B_S\, (1+\omega\, p_H\, (-1+p_M))-C_S\, p_M}$$

Equilibria/stability

We then solve for equilibria in the system (i.e., conditions under which both host and symbiont allele frequencies do not change, $\Delta p_H = \Delta p_M = 0$).

In[28]:=
```
Eqs = Solve[Δp_H == 0 && Δp_M == 0, {p_H, p_M}]
```

Solve::svars : Equations may not give solutions for all "solve" variables. >>

Out[28]= $\{\{p_M \to 1\}, \{p_H \to 0, p_M \to 0\}, \{p_H \to 1, p_M \to 0\}\}$

None of these are internal equilibria, which demonstrates that selective dynamics modeled with this system of equations do not result in stable maintenance of variation for either the host or symbiont.

To evaluate the stability of these equilibria, we calculate the Jacobean matrix for the system, based on the partial differentiation of each dynamic equation with respect to p_H and p_M.

In[29]:=
```
Jac = {{D[Δp_H, p_H], D[Δp_M, p_H]}, {D[Δp_H, p_M], D[Δp_M, p_M]}};
```

We then substitute each set of equilibrium conditions into the Jacobean to calculate the eigenvalues at that multivariable equilibrium point.

In[30]:=
```
FullSimplify[Eigenvalues[Jac /. Eqs[[1]]]]
```

Out[30]= $\left\{0, \frac{C_S - \omega B_S p_H}{1 + B_S - C_S}\right\}$

In[36]:=
```
FullSimplify[Eigenvalues[Jac /. Eqs[[2]]]]
```

Out[36]= $\left\{-\frac{\omega C_H}{-1 + C_H}, -\frac{C_S}{1 + B_S}\right\}$

This equilibrium, where $p_H = p_M = 0$, is only locally stable if $C_H > 1$, which is outside the range we consider. (It would potentially result in nonsensical fitness values.)

In[37]:=
```
FullSimplify[Eigenvalues[Jac /. Eqs[[3]]]]
```

Out[37]= $\left\{-\frac{\omega C_H}{1 + (-1 + \omega) C_H}, \frac{-\omega B_S + C_S}{-1 + (-1 + \omega) B_S}\right\}$

This equilibrium, where $p_H = 1$ and $p_M = 0$, is locally stable when $\omega B_S < C_S$. When $\omega B_S > C_S$, p_M increases until cooperative symbionts fix.

In[73]:=
```
VectorPlot[{ΔpH, ΔpM} /. {CH → 0.25, BH → 0.5, CS → 0.25, BS → 0.5, ω → 0.75},
  {pH, 0, 1}, {pM, 0, 1}, FrameLabel -> {pH, pM}]
```

Out[73]=